

UTILIZAR O QBASIC PARA APRENDER FÍSICA

Numa altura em que há já tantos e tão sofisticados programas de computador, desde folhas de cálculo como o EXCEL até programas tão fáceis de utilizar e tão poderosos como o INTERACTIVE PHYSICS, pode parecer algo estranho utilizar uma linguagem de programação para analisar fenómenos físicos e explorar modelos matemáticos. Uma linguagem de programação, como o QBASIC, apresenta, no entanto, várias vantagens na exploração de certos modelos matemáticos de situações físicas, nomeadamente: é fornecido com *todos* os computadores IBM e compatíveis IBM que utilizam o sistema operativo MS-DOS; permite a escrita do modelo matemático de uma forma muito próxima da normalmente utilizada; e permite o controlo do modelo e dos parâmetros do modelo de um modo directo e fácil.

A utilização do QBASIC para aprender Física pode ser feita a partir do 10.º ano de escolaridade, uma vez que não exige conhecimentos matemáticos muito aprofundados. Em alguns dos exemplos a seguir apresentados, recorre-se a um método numérico muito simples, compreensível pela maior parte dos alunos do 10.º ano.

Com um pequeno número de instruções do QBASIC (não mais do que 1%!) é possível explorar um grande número de conceitos físicos, como se procura mostrar neste artigo através de dois exemplos.

O que é o QBASIC

O QBASIC é uma versão reduzida da linguagem de programação Quick BASIC da Microsoft, sendo instalado automaticamente na directoria DOS do disco duro,

A secção "Aprender Física... com o computador" é coordenada por Vitor Duarte Teodoro e está aberta à colaboração de todos os leitores da Gazeta. Qualquer contribuição, que não deverá exceder duas páginas da revista, poderá ser enviada para

Vitor Duarte Teodoro
Faculdade de Ciências e Tecnologia (UNL)
2825 Monte de Caparica

Para entrar em contacto com o coordenador da secção, pode escrever para a morada acima indicada, utilizar o fax (01-2941005), o tel. (01-2954464, ext. 0410) ou, preferencialmente, o correio electrónico (vdt@mail.fct.unl.pt).

Todos os artigos não assinados são da responsabilidade do coordenador.

aquando da instalação do sistema operativo. Se a versão do sistema operativo for a versão em português, os menus do QBASIC surgem em português, bem como o respectivo «help».

Para executar o QBASIC, com o cursor no drive do disco duro (C:>), faz-se «cd dos» e escreve-se «qbasic».

Uma vez escrito um programa no editor do QBASIC (um programa é uma sequência de instruções que a linguagem de programação executa), este pode ser "corrido" utilizando a opção «Start» (ou «Começar», na versão em português) do menu «Run» (ou «Executar»).

A estrutura de um programa simples em QBASIC

Nas duas situações da mecânica clássica a seguir apresentadas, o programa escrito em QBASIC tem a seguinte estrutura:

- define-se um conjunto de parâmetros do modelo matemático que se vai utilizar;
- desenha-se um gráfico (ou um referencial), numa escala adequada;
- faz-se uma série de cálculos repetitivos no «ciclo FOR ... NEXT», na variável t , que toma valores entre $t = 0$ e $t = t_{\max}$.

A «Física» está «toda» nesta última parte do programa. É no *ciclo* que se calculam os valores das diversas variáveis, nos instantes sucessivos.

Nos dois exemplos apresentados, o cálculo dos valores das grandezas nos sucessivos intervalos de tempo faz-se de duas maneiras diferentes. Na primeira, os valores são calculados a partir de *funções* da variável tempo. No segundo caso, os valores são calculados a partir da determinação de *pequenas variações* das grandezas num intervalo de tempo muito pequeno, que é representado nos programas por dt . O valor de dt deve ser tão pequeno quanto necessário, mas convém não ser inferior a 0.001 para evitar certos erros de cálculo próprios dos computadores.

Investigar a trajectória de uma bola lançada para o ar na vertical, sendo desprezável a resistência do ar

Neste primeiro exemplo, mostra-se a trajectória num referencial Oxy de uma bola lançada verticalmente, com velocidade inicial de valor 20 m/s, a partir da posição de coordenadas $x = 10$ m e $y = 0$ m.

Na primeira parte do programa definem-se os parâmetros físicos necessários à simulação, bem como as escalas dos eixos, o tempo máximo, o tamanho da bola, etc.

Na segunda parte desenha-se um referencial no ecrã. A instrução «SCREEN 12» significa que o ecrã segue a

norma VGA (ecrã de 640 por 480 pontos). A instrução «VIEW (100, 100)-(400, 400), branco, preto» desenha uma «janela» no ecrã, entre os pontos de coordenadas (100, 100) e (400, 400). A instrução WINDOW (0, 0)-(escalax, escalay) define as escalas nesta janela, que vai funcionar como referencial. As instruções seguintes nesta parte do programa escrevem o nome dos eixos e o valor máximo das escalas.

No ciclo «FOR ... NEXT» determinam-se as posições nas coordenadas x e y , com passo («STEP») dt — isto é, de 0.1 s em 0.1, que foi o valor definido para dt . A instrução «CIRCLE (x, y), RaioDaBola, preto» desenha uma circunferência com centro nas coordenadas (x,y) e raio igual ao valor do parâmetro «RaioDaBola», na cor «preto». A linha «FOR i = 0 TO 1000: NEXT i: REM (pausa durante um certo intervalo de tempo)» é, simplesmente, um modo de «fazer o computador demorar um certo tempo até desenharmos a circunferência seguinte» — se se pretender aumentar ou diminuir este intervalo de tempo, aumenta-se ou diminui-se o valor máximo da variável «i» neste ciclo. Com a instrução seguinte, «CIRCLE (x, y), RaioDaBola, branco», apaga-se a circunferência desenhada a preto. Finalmente, nas três linhas seguintes, escreve-se os valores de tempo e das coordenadas, no topo do ecrã.

A primeira das figuras junto, mostra o ecrã no instante $t = 2,0$ s, quando se atinge a altura máxima.

Se se pretender uma representação estroboscópica, apaga-se a linha «CIRCLE (x, y), RaioDaBola, branco» (ou desactiva-se a linha colocando REM no seu início). Obtém-se, então, um ecrã como o da segunda figura.

Facilmente se pode efectuar a análise pormenorizada deste modelo, mudando os valores dos parâmetros no início do programa: que sucede se se aumentar a velocidade inicial? E se se partir de outra posição? Como transformar o modelo numa queda livre de uma certa altura? Que sucede se não houver aceleração da gravidade? Etc. Em alguns casos, pode ser necessário alterar as escalas dos eixos...

Sugere-se ao leitor que copie o programa para o seu computador e tente explorar as suas possibilidades didácticas.

```



REM lançamento de uma bola na vertical visto num referencial Oxy

REM parametros
x0 = 10
y0 = 0
v0y = 20
g = 10
ay = -g
dt = .1
tmax = 4
escalax = 30
escalay = 30
RaioDaBola = .2
preto = 0
branco = 7

REM referencial
SCREEN 12
VIEW (100, 100)-(400, 400), branco, preto
WINDOW (0, 0)-(escalax, escalay)
LOCATE 28, 40: PRINT "x"
LOCATE 12, 5: PRINT "y"
LOCATE 27, 12: PRINT 0: LOCATE 27, 49: PRINT escalax
LOCATE 25, 10: PRINT 0: LOCATE 7, 9: PRINT escalay

FOR t = 0 TO tmax STEP dt
  x = x0
  y = y0 + v0y * t + (1 / 2) * ay * t ^ 2
  CIRCLE (x, y), RaioDaBola, preto
  FOR i = 0 TO 1000: NEXT i: REM(pausa durante um certo int. tempo)
  CIRCLE (x, y), RaioDaBola, branco
  LOCATE 2, 25: PRINT USING "t = ##.## "; t
  LOCATE 3, 25: PRINT USING "x = ##.## "; x
  LOCATE 4, 25: PRINT USING "y = ##.## "; y
NEXT t

```



Estudar um oscilador no laboratório... e no computador

O estudo dos osciladores é um tópico que, aparentemente, exige conhecimentos matemáticos avançados. Vamos apresentar uma actividade relativamente simples envolvendo a determinação do período de um oscilador, primeiro com um cronómetro e depois com o modelo matemático do oscilador.

Utiliza-se uma das várias molas que existem nos laboratórios escolares. Com pesos adequados e uma régua, determina-se a *constante da mola* em newtons por metro. Para uma certa mola, obteve-se $k = 2,9 \text{ N/m}$.

Coloca-se a mola a oscilar, suspendendo um corpo de 100 g, por exemplo, e esticando-a até um certo valor. Esse valor é a ordenada mínima ($y = -0,13$; a origem é o ponto de equilíbrio, com o corpo suspenso).

A componente, segundo o eixo dos yy , da força elástica na mola é $F_{\text{elástica}} = -ky = -2,9y$. A esta força há que somar o valor de 1,0 N, a força exercida pela mola na posição de equilíbrio — ver esquema.

Além da força elástica e da força da mola, há a considerar a força gravítica sobre o corpo, cuja componente segundo o eixo vertical é: $F_{\text{grav}} = -mg = -0,100 \times 10 = 1,0 \text{ N}$.

No programa ao lado, no ciclo «FOR ... NEXT», depois de se calcularem estas componentes das duas forças que actuam no oscilador, calcula-se a componente da resultante das forças e utiliza-se esse valor para calcular a aceleração nesse instante.

Em seguida, calcula-se a velocidade no instante seguinte, admitindo que é igual à soma da velocidade no instante anterior com a *variação* de velocidade num *pequeno* intervalo de tempo dt , que é dada pelo produto $ay * dt$ — a aceleração *mede* a variação de velocidade em cada *instante*.

Por fim, calcula-se a coordenada y no instante seguinte, admitindo que é igual à soma de y no instante anterior com a *variação* de ordenada num *pequeno* intervalo de tempo dt , que é dada pelo produto $vy * dt$ — a velocidade *mede* a variação de posição em cada *instante*.

Estes cálculos repetem-se em intervalos de tempo sucessivos dt em dt . Quanto mais pequeno for este intervalo de tempo, melhor é, em princípio, a aproximação conseguida por este processo. Cuidado, no entanto, com os dt muito pequenos...

Utilizando o gráfico, é possível verificar se o valor que se obtém para o período do oscilador está de acordo com o valor determinado experimentalmente — se a experiência for feita com cuidado, a diferença não excede os 5%! Pode controlar-se o valor do parâmetro «tmax» para determinar o período de oscilação, a partir do gráfico.

```

REM simulacao do movimento de um oscilador

REM parametros
y = -.13
vy = 0
m = .1
g = 10
k = 2.9
dt = .001
tmax = 10
escalay = .5

REM Grafico e calculo
SCREEN 12
VIEW (100, 100)-(600, 400), 6, 2
WINDOW (0, -escalay)-(tmax, escalay)
LOCATE 28, 50: PRINT "t/s"
LOCATE 12, 5: PRINT "y/m"
LOCATE 27, 12: PRINT 0: LOCATE 27, 61: PRINT tmax
LOCATE 25, 9: PRINT -escalay: LOCATE 7, 9: PRINT escalay
FOR t = 0 TO tmax STEP dt
  Felastica = k * y
  Fmola = m * g
  Fgrav = -m * g
  Fresultante = Felastica + Fgrav + Fmola
  ay = Fresultante / m
  vy = vy + ay * dt
  y = y + vy * dt
  PSET (t, y), 0
  LOCATE 2, 25: PRINT USING "t = #.#"; t
  LOCATE 3, 25: PRINT USING "y = #.##"; y
  LOCATE 4, 25: PRINT USING "vy = #.##"; vy
NEXT t
    
```

